

Contents

1	Note on this document	1
2	Calculation of axion mass range of a helioscope	1
2.1	Analytical formula for effective photon mass	6
2.2	Upper mass range for vacuum	6
2.3	Calculating axion conversion probability	7
2.4	Find our error in the calculation	23
2.5	Absorption of X-rays in helium	24
2.5.1	Absorption in beam line at room temp @ 1 bar pressure	25
2.6	First pressure value for He filling	26
2.6.1	Full vacuum sensitivity curve	27
2.6.2	Determine the next pressure (density) step	28
3	Notes	36
3.1	TODO Evaluate if this m_γ results in the same value as our fn	36
3.2	Momentum transfer units	36
3.3	Attenuation length, momentum transfer and length	37
4	Appendix	appendix 38

1 Note on this document

In order to run this code you obviously need Nim in your `PATH` as well as all packages, which are imported. In addition you need `ntangle` in your path. With those in place, run:

```
1 ntangle axionMass.org
2 nim c -r axionMass.nim
```

and the code is extracted and run, regenerating all plots.

2 Calculation of axion mass range of a helioscope

For BabyIAXO it would be nice to know:

1. the upper axion mass range for usage with vacuum
2. the resonant axion mass for a specific helium density

Starting point for this discussion is the IAXO gas phase study by Biljana and Kresimir. The coherence condition for axions is

$$qL < \pi \quad (1)$$

$$\text{where } q = \frac{m_a^2}{2E_a} \quad (2)$$

with L the length of the magnetic field (20m for IAXO, 10m for BabyIAXO), m_a the axion mass and E_a the axion energy (taken from solar axion spectrum).

In the presence of a low pressure gas, the photon receives an effective mass m_γ , resulting in a new q :

$$q = \left| \frac{m_\gamma^2 - m_a^2}{2E_a} \right| \quad (3)$$

Thus, we first need some values for the effective photon mass in a low pressure gas, preferably helium.

From this we can see that coherence in the gas is restored if $m_\gamma = m_a$, $q \rightarrow 0$ for $m_a \rightarrow m_\gamma$. This means that in those cases the energy of the incoming axion is irrelevant for the sensitivity!

In order to calculate some values, we'll write some Nim code to calculate and plot the dependence of m_a on the gas pressure.

First we have to import some modules we'll need:

```
1 import sequtils, seqmath, ggplotnim, strformat, algorithm, nlopt, options, strutils
```

An approximation for the dependence of m_γ on the surrounding gas, was found in: <https://www.sciencedirect.com/science/article/pii/S0370269302018221> and can be written as:

$$m_\gamma = \sqrt{\frac{4\pi\alpha n_e}{m_e}} \quad (4)$$

where α is the fine structure constant, m_e the electron mass and n_e the electron number density. However, this equation still has to be fixed regarding the units. The units as such are:

$$\text{eV} = \sqrt{\frac{1}{\text{eVm}^3}} \quad (5)$$

We can fix this, by replacing the m^3 by their natural unit equivalents in eV: $1 \text{ eV}^{-1} == 1.97\text{e-}7 \text{ m}$. That is, replace the $1 / \text{m}^3$ by $1.97\text{e-}7 \text{ eV}^3$, for a final equation:

$$m_\gamma = \sqrt{\frac{4\pi\alpha n_e \cdot (1.97\text{e-}7)^3}{m_e}} \quad (6)$$

A proc to calculate an effective mass from an electron number density is thus:

```
1 proc effPhotonMass(ne: float): float =
2   ## returns the effective photon mass for a given electron number density
3   const alpha = 1.0 / 137.0
4   const me = 511e3 # 511 keV
5   # note the 1.97e-7 cubed to account for the length scale in `ne`
6   result = sqrt( pow(1.97e-7, 3) * 4 * PI * alpha * ne / me )
```

This means we need to calculate the electron number density n_e for a given gas. For practical reasons it's probably easier to calculate it from a molar density in mol / m^3 for a gas (see here):

$$n_e = Z \cdot N_A \cdot c \quad (7)$$

where N_A is the Avogadro constant ($N_A = 6.022\text{e}23 \text{ mol}^{-1}$) and c the molar density of the gas in mol / m^3 and Z appears since we have Z electrons per molecule in the gas.

```
1 proc numDensity(c: float): float =
2   ## converts a molar concentration in mol / m^3 to a number density
3   const Z = 2 # number of electron in helium atom
4   result = Z * 6.022e23 * c
```

The standard atomic weight of helium is $A_r = 4.002602$, resulting in a molar mass of $M(\text{He}) = 4.002602 \text{ g mol}^{-1}$.

Given the ideal gas law:

$$PV = nRT \quad (8)$$

with the pressure P , the gas volume V , the amount of gas in mol n , the universal gas constant R : $R = 8.314 \text{ J K}^{-1} \text{ mol}^{-1}$, and the temperature T .

So to calculate the molar gas amount from a pressure, volume and temperature:

```
1 proc molarAmount(p, vol, temp: float): float =
2   ## calculates the molar amount of gas given a certain pressure,
3   ## volume and temperature
4   ## the pressure is assumed in mbar
5   const gasConstant = 8.314 # joule K^-1 mol^-1
6   let pressure = p * 1e2 # pressure in Pa
7   result = pressure * vol / (gasConstant * temp)
8   #echo "Molar amount for P = ", pressure, " Pa is ", result
```

We know the volume and temperature and want to calculate the dependency of m_a on different pressure values. Thus, we can calculate n for each pressure and insert it into the number density proc to calculate the effective photon mass.

As mentioned above, coherence in the gas is restored if $m_\gamma = m_a$. So all we have to do is to calculate m_γ for different pressures and then we know the effective axion mass we're sensitive to in BabyIAXO given a certain helium pressure.

So calculate the effective photon mass for a given pressure in BabyIAXO:

```
1 proc babyIaxoEffMass(p: float): float =
2   ## calculates the effective photon (and thus axion mass) for BabyIAXO given
3   ## a certain helium pressure in the BabyIAXO magnet
4   const vol = 10.0 * (PI * pow(0.3, 2)) # 10m length, bore radius 30 cm
5   # UPDATE: IAXO will be run at 4.2 K instead of 1.7 K
6   # const temp = 1.7 # assume 1.7 K same as CAST
7   const temp = 4.2
8   once:
9     echo "BabyIAXO magnet volume is ", vol, " m^3"
10    echo "BabyIAXO magnet temperature is ", temp, " K"
11    let amountMol = molarAmount(p, vol, temp) # amount of gas in mol
12    let numPerMol = numDensity(amountMol / vol) # number of electrons per m^3
13    #echo "Num electrons per m^3 ", numPerMol
14    result = effPhotonMass(numPerMol)
```

All that is left to do then is to create a set of pressure values that we want to plot against and calculate the sensitive axion mass for those values for BabyIAXO. Since we probably want to cover a log space, create a helper proc `logspace` from `linspace`:

```
1 proc logspace(start, stop: float, num: int, base = 10.0): seq[float] =
2   ## generates evenly spaced points between start and stop in log space
3   let linear = linspace(start, stop, num)
4   result = linear.mapIt(pow(base, it))
```

So generate values and create a plot. Let's write a proc to generate a plot given some mbar range and a plot filename:

```
1 proc makePlot(pstart, pstop: float, fname: string, log = false) =
2   let pressures = logspace(pstart, pstop, 1000) # 1000 values from 1e-5 mbar to 500 mbar
3   let masses = pressures.mapIt(babyIaxoEffMass(it)) # corresponding masses
4   # convert both seqs to a dataframe
5   let df = seqsToDf({"P / mbar" : pressures, "m_a / eV" : masses})
6   let plt = ggplot(df, aes("P / mbar", "m_a / eV")) +
7     geom_line() +
8     ggtitle("Sensitive axion mass in eV depending on helium pressure in mbar")
9   if not log:
10    plt + ggsave(fname)
11  else:
12    plt + scale_x_log10() + scale_y_log10() + ggsave(fname)
```

First for a logspace from 1e-6 to 1e2 mbar:

```
1 makePlot(-6.0, 2.0, "axionMassesFullRange.pdf")
```

and now for a low pressure range:

```
1 makePlot(-6.0, -2.0, "axionMassesZoomed.pdf")
```

and finally as a log-log plot in the full range.

```
1 makePlot(-6.0, 2.0, "axionMassesLogLog.pdf", log = true)
```

These plots are shown in fig. 1 and 2 and the log-log plot shown in 3.




Figure 1: Axion masses in eV depending on the pressure in BabyIAXO in mbar in the full range from 10×10^{-6} mbar to 10×10^2 mbar.




Figure 2: Axion masses in eV depending on the pressure in BabyIAXO in mbar in a zoomed range from 10×10^{-6} mbar to 10×10^{-2} mbar.




Figure 3: Axion masses in eV depending on the pressure in BabyIAXO in mbar in the full range from 10×10^{-6} mbar to 10×10^2 mbar as a log-log plot.

2.1 Analytical formula for effective photon mass

UPDATE: updated to fix issues from 2.4.

We can easily derive an analytical expression for the above calculation, by simply inserting all equations into one another.

$$m_\gamma = \sqrt{\frac{(1.97 \times 10^{-7})^3 4\pi\alpha Z N_a 100 \cdot P(\text{mbar})}{m_e R T}} \quad (9)$$

if the pressure is given in mbar (factor 100 accounts for that).

which after inserting all numbers and multiplying them gives the following expression with:

- $Z = 2$
- $T = 4.2 \text{ K}$
- $R = 8.314 \text{ J K}^{-1} \text{ mol}^{-1}$
- $N_A = 6.022 \times 10^{23} \text{ mol}^{-1}$
- $\alpha = 1 / 137$
- $m_e = 511,000 \text{ eV}$

which results in:

$$m_\gamma = 1.94081 \times 10^{-2} \cdot \sqrt{4\pi P(\text{mbar})} \quad (10)$$

Let's define this as a function and check that the resulting values are actually the same as for the above calculation:

```

1  proc analyticalCalc(p: float): float =
2      ## calculates the effective mass for a given pressure from a single equation
3      ## derived from inserting all above into one
4      result = 1.94081e-2 * sqrt(4 * PI * p )
5
6  block:
7      let pressures = logspace(-6.0, 2.0, 1000)
8      let massesNum = pressures.mapIt(babyIaxoEffMass(it))
9      let massesAna = pressures.mapIt(analyticalCalc(it))
10     #echo massesNum
11     #echo massesAna
12     func almostEqual(a, b: seq[float]): bool = zip(a, b).mapIt(abs(it[0] - it[1]) < 1e-5).allIt(it)
13     doAssert massesNum.almostEqual(massesAna)

```

Which is indeed the case (as the assertion holds during runtime).

2.2 Upper mass range for vacuum

Finally also calculate the mass range one is sensitive to, if the magnet is not filled with helium.

In those cases, the limit is just given by the coherence condition mentioned at the top of the file $qL < \pi$. Using the boundary as the condition to find m_a we have:

$$qL = \pi \quad (11)$$

$$\frac{m_a^2}{2E_a} L = \pi \quad (12)$$

$$m_a = \sqrt{\frac{\pi 2E_a}{L}} \quad (13)$$

where we again have to add a factor of $1.97 \times 10^{-7} \text{ m}$ for the length scale L :

$$m_a = \sqrt{\frac{\pi 2E_a \cdot 1.97 \times 10^{-7}}{L}}$$

Note that in this case the effective mass one is sensitive to is directly dependent on the axion's energy, in contrast to the resonant case in the presence of an effective photon mass.

So we can calculate the mass limit in dependence of the axion's energy as:

```
1 proc vacuumMassLimit(energy: float, magnetLength = 10.0): float =
2   ## given an axion energy in keV, calculate the limit of coherence
3   ## for the vacuum case in BabyIAXO
4   # note the length scale 1.97e-7 to take into account the meter scale in
5   # babyIaxoLen
6   let babyIaxoLen = magnetLength / 1.97e-7 # length in "eV"
7   result = sqrt(PI * 2 * energy * 1e3 / babyIaxoLen) # convert energy to eV
```

So let's also create a plot covering the interesting energy range from 0 to 10 keV:

```
1 let energies = linspace(0.0, 10.0 , 1000) # from 0 to 10 keV
2 let massLimits = energies.mapIt(vacuumMassLimit(it))
3 let df = seqsToDf({"E / keV" : energies, "m_a / eV" : massLimits})
4 ggplot(df, aes("E / keV", "m_a / eV")) +
5   geom_line() +
6   ggtitle("Sensitive axion mass limit in eV for BabyIAXO in vacuum run") +
7   ggsave("vacuumMassLimit.pdf")
```

Now cross check the value for the vacuum mass limit with the value calculated by Biljana and Kreso for IAXO, which was $m_a < 1.62e-2$ eV for IAXO (length 20m) at an energy of 4.2 keV:

```
1 let iaxoLimit = vacuumMassLimit(4.2, 20.0)
2 func almostEqual(a, b: float, eps = 1e-3): bool = abs(a - b) < eps
3 doAssert iaxoLimit.almostEqual(1.61e-2) # value from IAXO gas phase study
4 echo "Full IAXO mass limit @ 4.2 keV = ", iaxoLimit, " eV"
```

And we see that this is indeed the case!

Compared with that the value of the vacuum limit at 4.2 keV for BabyIAXO is:

```
1 const babyIaxoVacuumMassLimit = vacuumMassLimit(4.2)
2 echo "BabyIAXO mass limit @ 4.2 keV = ", babyIaxoVacuumMassLimit, " eV"
```

2.3 Calculating axion conversion probability

Now we will try to calculate the axion conversion probability in Helium at a given gas density and temperature.

The axion-photon conversion probability $P_{a \rightarrow \gamma}$ in general is given by:

$$P_{a \rightarrow \gamma} = \left(\frac{g_{a\gamma} B}{2} \right)^2 \frac{1}{q^2 + \Gamma^2/4} \left[1 + e^{-\Gamma L} - 2e^{-\frac{\Gamma L}{2}} \cos(qL) \right], \quad (14)$$

where Γ is the inverse absorption length for photons (or attenuation length).

Aside from Γ we have all values at hand. At the end we want to have $P_{a \rightarrow \gamma}$ as a function of $P_{a \rightarrow \gamma}(\Gamma, m_a)$, i.e. we pick a specific B and $g_{a\gamma}$, set L and wish to evaluate the conversion probability based on Γ (depends on the gas pressure) and the axion mass. That way we can calculate the conversion probability FWHM at a given pressure.

Following Theodoro's PhD thesis (see 3 below), the attenuation length can be calculated from:

$$\Gamma = \rho \left(\frac{\mu}{\rho} \right), \quad (15)$$

where μ/ρ is the so called mass attenuation coefficient. The values for the mass attenuation coefficient are tabulated, e.g. by NIST: <https://physics.nist.gov/PhysRefData/XrayMassCoef/tab3.html> Since

Energy MeV	μ/ρ $\text{cm}^2 \text{g}^{-1}$	μ_{en}/ρ $\text{cm}^2 \text{g}^{-1}$
1.00000 (-03)	6.084 (+01)	6.045 (+01)
1.50000 (-03)	1.676 (+01)	1.638 (+01)
2.00000 (-03)	6.863 (+00)	6.503 (+00)
3.00000 (-03)	2.007 (+00)	1.681 (+00)
4.00000 (-03)	9.329 (-01)	6.379 (-01)
5.00000 (-03)	5.766 (-01)	3.061 (-01)
6.00000 (-03)	4.195 (-01)	1.671 (-01)
8.00000 (-03)	2.933 (-01)	6.446 (-02)
1.00000 (-02)	2.476 (-01)	3.260 (-02)
1.50000 (-02)	2.092 (-01)	1.246 (-02)
2.00000 (-02)	1.960 (-01)	9.410 (-03)
3.00000 (-02)	1.838 (-01)	1.003 (-02)
4.00000 (-02)	1.763 (-01)	1.190 (-02)
5.00000 (-02)	1.703 (-01)	1.375 (-02)
6.00000 (-02)	1.651 (-01)	1.544 (-02)
8.00000 (-02)	1.562 (-01)	1.826 (-02)
1.00000 (-01)	1.486 (-01)	2.047 (-02)
1.50000 (-01)	1.336 (-01)	2.424 (-02)
2.00000 (-01)	1.224 (-01)	2.647 (-02)
3.00000 (-01)	1.064 (-01)	2.868 (-02)
4.00000 (-01)	9.535 (-02)	2.951 (-02)
5.00000 (-01)	8.707 (-02)	2.971 (-02)
6.00000 (-01)	8.054 (-02)	2.959 (-02)
8.00000 (-01)	7.076 (-02)	2.890 (-02)
1.00000 (+00)	6.362 (-02)	2.797 (-02)
1.25000 (+00)	5.688 (-02)	2.674 (-02)
1.50000 (+00)	5.173 (-02)	2.555 (-02)
2.00000 (+00)	4.422 (-02)	2.343 (-02)
3.00000 (+00)	3.503 (-02)	2.019 (-02)
4.00000 (+00)	2.949 (-02)	1.790 (-02)
5.00000 (+00)	2.577 (-02)	1.622 (-02)
6.00000 (+00)	2.307 (-02)	1.493 (-02)
8.00000 (+00)	1.940 (-02)	1.308 (-02)
1.00000 (+01)	1.703 (-02)	1.183 (-02)
1.50000 (+01)	1.363 (-02)	9.948 (-03)
2.00000 (+01)	1.183 (-02)	8.914 (-03)

Table 1: Table from NIST of the mass attenuation length of photons in helium. The right most column is the mass-energy absorption coefficient, which we do not care about here.

we mainly care about ^4He in the context of this document, we can take the helium entry straight from <https://physics.nist.gov/PhysRefData/XrayMassCoef/ElemTab/z02.html>, which conveniently lists these values in useful energy ranges, see tab. 1.

In order to evaluate the mass attenuation at any energy, we need an interpolation function. Conveniently Javier Galan (PhD thesis) fitted such a function for us already (taken from Theodoros' thesis),

$$\log\left(\frac{\mu}{\rho}\right)(E) = 1.5832 + 5.9195e^{-0.353808E} + 4.03598e^{-0.970557E} \quad (16)$$

We will now define this function in our Nim program and generate the datapoints for the interpolation in our energy range we consider above:

```
1 proc logMassAttenuation(e: float): float =
2   ## calculates the logarithm of the mass attenuation coefficient for a given
3   ## energy `e` in `keV` and the result in `cm^2/g`
4   result = -1.5832 + 5.9195 * exp(-0.353808 * e) + 4.03598 * exp(-0.970557 * e)
5
6 let logMuOverRho = energies.mapIt(logMassAttenuation(it))
7 # now the non-log values
8 let muOverRho = logMuOverRho.mapIt(exp(it))
```

Before we can plot this together with the tabulated data, we have to parse the table above. The raw data is also stored in `mass_attenuation_nist_data.txt`. We will now parse it into a dataframe and then drop all values outside the range of the energies we consider:

```
1 const massAttenuationFile = "mass_attenuation_nist_data.txt"
2 # skip one line after header, second header line
3 var dfMuRhoTab = toDf(readCsv(massAttenuationFile, skipLines = 1, sep = ' '))
4 # convert MeV energy to keV
5 .mutate(f{"Energy" ~ "Energy" * 1000.0})
6 .filter(f{"Energy" >= energies.min and "Energy" <= energies.max})
```

which leaves us with the following table of values tab. 2:

Idx	Energy / keV	mu/rho	mu _{en} /rho
0	1	60.84	60.45
1	1.5	16.76	16.38
2	2	6.863	6.503
3	3	2.007	1.681
4	4	0.9329	0.6379
5	5	0.5766	0.3061
6	6	0.4195	0.1671
7	8	0.2933	0.06446
8	10	0.2476	0.0326

Table 2: All mass attenuation coefficients (μ/ρ) in the range of our energies considered 0 keV to 10 keV

Now we can finally compare the interpolation function with the real coefficients and see if they actually match:

```
1 # create df of interpolated values
2 let dfMuRhoInterp = seqsToDf({ "E / keV" : energies,
3                                "mu/rho" : muOverRho,
4                                "log(mu/rho)" : logMuOverRho})
5 # rename the columns of the tabulated values df and create a log column
6 dfMuRhoTab = dfMuRhoTab.rename(f{"E / keV" ~ "Energy"})
7 .mutate(f{"log(mu/rho)" ~ ln("mu/rho")})
8 # build combined DF
```

```

9  let dfMuRho = bind_rows(["Interpolation", dfMuRhoInterp),
10                          ("NIST", dfMuRhoTab)],
11                          id = "type")
12  echo dfMuRho
13  # create plot of log values
14  ggplot(dfMuRho, aes("E / keV", "log(mu/rho)", color = "type")) +
15    geom_line(data = dfMuRho.filter(f{"type" == "Interpolation"})) +
16    geom_point(data = dfMuRho.filter(f{"type" == "NIST"})) +
17    ggtitle("Mass attenuation coefficient interpolation and data") +
18    ggsave("log_mass_attenuation_function.pdf")
19
20  # and the plot of the raw mu/rho values
21  ggplot(dfMuRho, aes("E / keV", "mu/rho", color = "type")) +
22    geom_line(data = dfMuRho.filter(f{"type" == "Interpolation"})) +
23    geom_point(data = dfMuRho.filter(f{"type" == "NIST"})) +
24    scale_y_log10() +
25    ggtitle("Mass attenuation coefficient interpolation and data") +
26    ggsave("mass_attenuation_function.pdf")

```

The figure of the mass attenuation function (non-log data) is shown in fig. 4.




Figure 4: Mass attenuation coefficient comparison of interpolation function with the tabulated data from NIST for ^4He .

Now all that remains is to put the interpolation function to use, as it apparently correctly describes the NIST data. The only sanity check left before we do that is to check whether the effective mass calculation of Theodoros' thesis matches with the function we obtained.

Our simplified formula ended up being:

$$m_\gamma = 1.37236 \times 10^{-2} \cdot \sqrt{4\pi P} \quad (17)$$

while according to equation 7.3:

$$m_\gamma = 28.77 \sqrt{\frac{Z}{A}} \rho \quad (18)$$

where for ^4He $Z = 2$ and $A \sim 4 \text{ g mol}^{-1}$.

The ideal gas law in molar form is:

$$P = \rho \frac{R}{M} T, \quad (19)$$

where M is the molar mass of our gas. Since we assume it's pure ^4He , this is $M = 4 \text{ g mol}^{-1}$. However, if we simply replace the pressure in our equation by this equation, the result will be still dependent both on T , which is because equation 17 already includes our temperature of $T = 4.2 \text{ K}$ (as well as R as a matter of fact). Let's fix that:

$$m_\gamma = 1.37236 \times 10^{-2} \cdot 4.2 \cdot 8.314 \sqrt{\frac{4\pi P}{RT}} \quad (20)$$

$$m_\gamma = 0.47921 \sqrt{\frac{4\pi P}{RT}}. \quad (21)$$

Now we can finally replace P :

$$m_\gamma = 0.47921 \sqrt{\frac{4\pi \rho R T}{M T}} \quad (22)$$

$$m_\gamma = 0.47921 \sqrt{\frac{4\pi \rho}{M}} \quad (23)$$

and inserting the molar mass of helium $M(\text{He}) = 4.002602 \text{ g mol}^{-1}$ yields:

$$m_\gamma = 0.8491 \sqrt{\rho} \quad (24)$$

and comparing that to eq. 18 by inserting Z and A :

$$m_\gamma = 28.77 \sqrt{\frac{2}{4.002602} \rho} \quad (25)$$

$$m_\gamma = 20.33685 \sqrt{\rho} \quad (26)$$

which, *ehm* does not match.

Why?

Will figure out the error later. For now continue on with the actual calculation of the conversion probability.

If we take one of our effective mass calculations for granted for the moment, we can calculate the probability as we defined it further above, eq. 14.

```

1  proc momentumTransfer(m_gamma, m_a: float, E_a = 4.2): float =
2      ## calculates the momentum transfer for a given effective photon
3      ## mass `m_gamma` and axion mass `m_a` at an axion energy of
4      ## 4.2 keV `E_a` (by default).
5      #const c = 299792458
6      result = abs((m_gamma * m_gamma - m_a * m_a) / (2 * E_a * 1000.0))
7
8  proc axionConversionProb(m_a, m_gamma, gamma: float, length = 10.0): float =
9      ## given an axion mass and an inverse absorption length
10     ## calculates the conversion probability with BabyIAXOs magnet
11     ## properties. Axion coupling constant taken to be `1` +1e-11+.
12     ## By default uses BabyIAXO length of `10m`
13     # both `g_agamma` and `B` only scale the absolute value `P`, does not matter
14     const g_agamma = 1.0 #1e-11
15     const B = 4.5 # T, actually don't know the real number right now
16     # convert length in `m` to `eV`
17     let L = length / 1.97e-7 # m

```

```

18 let q = momentumTransfer(m_gamma, m_a)
19 let term1 = pow(g_agamma * B / 2.0, 2)
20 let term2 = 1.0 / (q * q + gamma * gamma / 4)
21 let term3 = 1.0 + exp(-gamma * L) - 2 * exp(-gamma * L / 2) * cos(q * L)
22 result = term1 * term2 * term3

```

Now all that is left to do is think up some reasonable numbers for the axion mass we want to investigate, calculate the effective photon masses for all pressures and the mass attenuation values and we're done. Start with effective photon masses,

```

1 let pressures = logspace(-6.0, 2.0, 4) # take only few values for a start
2 let m_gamma = pressures.mapIt(babyIaxoEffMass(it))

```

now for the mass attenuations, we need to derive a density from the pressure values we have:

```

1 proc density(p: float, temp = 4.2): float =
2   ## returns the density of the gas for the given pressure.
3   ## The pressure is assumed in `mbar` and the temperature (in `K`).
4   ## The default temperature corresponds to BabyIAXO aim.
5   ## Returns the density in `g / cm^3`
6   const gasConstant = 8.314 # joule K^-1 mol^-1
7   const M = 4.002602 # g / mol
8   let pressure = p * 1e2 # pressure in Pa
9   # factor 1000 for conversion of M in g / mol to kg / mol
10  result = pressure * M / (gasConstant * temp * 1000.0)
11  # convert the result to g / cm^3 for use with mass attenuations
12  result = result / 1000.0

```

Let's add a convenience proc to directly calculate the attenuation length from a pressure in mbar:

```

1 proc attenuationLength(p: float): float =
2   ## for a given pressure in `mbar` returns the attenuation length
3   ## `Γ` in units of `eV`.
4   # multiply by `100` to convert `Γ` from `1 / cm` to `1 / m`
5   # multiply by `1.97e-7` to convert `1 / m` to `1 / eV`
6   result = 1.97e-7 * 100.0 * density(p) * exp(logMassAttenuation(4.2))

```

So the densities are:

```

1 let densities = pressures.mapIt(density(it))

```

Let's first calculate the Γ values for a fixed energy:

```

1 let gammas = densities.mapIt(it * exp(logMassAttenuation(4.2)))

```

For each of the (Γ, q) pairs we will generate one plot, similar to fig. 5.

To generate it we still need a reasonable guess for the axion mass of interest. We can essentially use the effective photon mass as a baseline and generate N values around $O(1\%)$ of the nominal value. Let's try that:

```

1 proc genAxionConvPlot(gamma, m_gamma: float, nameSuffix = "",
2                       start = 0.0, stop = 0.0, length = 10.0) =
3   ## generates a single axion conversion probability plot
4   # calculate reasonable `m_a` values

```




Figure 5: Example of an axion conversion probability taken from the IAXO gas phase study.

```

5 echo "Gamma: ", gamma
6 echo "m_gamma: ", m_gamma
7 var m_a: seq[float]
8 if start != stop:
9   m_a = linspace(start, stop, 1000)
10 else:
11   m_a = linspace(m_gamma * 0.99, m_gamma * 1.01, 1000)
12
13 let qs = m_a.mapIt(momentumTransfer(m_gamma, it))
14 # plot momentum transfers by themselves
15 let dfMom = seqsToDf(m_a, qs)
16 ggplot(dfMom, aes("m_a", "qs")) +
17   geom_line() +
18   ggsave("momentumTransfers.pdf")
19
20 let prob = m_a.mapIt(axionConversionProb(it, m_gamma, gamma, length = length))
21 let df = seqsToDf({ "m_a / eV" : m_a,
22                   "P_a->gamma" : prob })
23 echo df
24 #echo df.summarize(f{"P_a->gamma" ~ min("P_a->gamma")})
25 #echo df.pretty(-1)
26 ggplot(df, aes("m_a / eV", "P_a->gamma")) +
27   geom_line() +
28   ggtitle(&"Axion conversion probability for  $\Gamma = \{\text{gamma} : .2\text{e}\}$ ,  $m_ = \{\text{m\_gamma} : .2\text{e}\}$ ") +
29   ggsave(&"axion_conversion_prob_{nameSuffix}.pdf")
30 if nameSuffix == "1":
31   writeFile("dfdata.txt", df.pretty(-1))

```

and call it for all values:

```

1 doAssert gammas.len == m_gammas.len
2 for i in 0 ..< gammas.len:
3   genAxionConvPlot(gammas[i], m_gammas[i], $i)
4 let dfctest = seqsToDf({ "gammas" : gammas,
5                         "m_gammas" : m_gammas })
6 ggplot(dfctest, aes("gammas", "m_gammas")) +
7   geom_line() +
8   ggsave("test_gamma.pdf")

```

Which currently is plain wrong.

Let's debug. **UPDATE:** see 2.4, the fixes from there were already applied to the code above.

To avoid any more mistakes, let's write some tests to check whether we can reproduce first the two

values we can read off from the IAXO gas phase study paper. Looking at both fig. 5 and the equivalent for a pressure equivalent of 3 bar at room temperature yields the following reference values, tab. 3.

We calculate the pressure corresponding to the room temperature pressures using the ideal gas law again:

```

1 proc pressureAtTemp(p: float, T = 4.2, T2 = 293.0): float =
2   ## converts a given pressure `p` at `T` to an equivalent pressure at `T2`
3   result = p * T2 / T
4
5 let p1bar = pressureAtTemp(1000.0, 293, 4.2)
6 let p3bar = pressureAtTemp(3000.0, 293, 4.2)
7 echo "Pressure for equivalent of 1 bar @ 293 K ", p1bar
8 echo "Pressure for equivalent of 3 bar @ 293 K ", p3bar
9 let m_gamma_1bar = babyIaxoEffMass(p1bar)
10 let m_gamma_3bar = babyIaxoEffMass(p3bar)
11 echo "m_gamma for pressure equivalent of 1 bar @ 293 K ", m_gamma_1_bar
12 echo "m_gamma for pressure equivalent of 3 bar @ 293 K ", m_gamma_3_bar
13 doAssert almostEqual(m_gamma_1_bar, 0.26)
14 doAssert almostEqual(m_gamma_3_bar, 0.4483, eps = 1e-2)

```

Which yields the values shown in the table.

P @ 293 K / bar	P @ 4.2 K / mbar	m_γ / eV
1.0	14.3345	0.26048
3.0	43.0034	0.45117

Table 3: Reference values extracted from the IAXO gas phase study paper.

Now we'll generate the exact plots for the conversion probability for these values.

```

1 let gamma_1bar = attenuationLength(p1bar)
2 let gamma_3bar = attenuationLength(p3bar)
3 genAxionConvPlot(gamma_1bar, m_gamma_1bar, "1bar_equiv")#, 0.2593, 0.2616
4 genAxionConvPlot(gamma_3bar, m_gamma_3bar, "3bar_equiv")#, 0.4485, 0.4535
5 # and now for reference the IAXO (~20m) plots
6 genAxionConvPlot(gamma_1bar, m_gamma_1bar, "1bar_equiv_20m", length = 20.0)
7 genAxionConvPlot(gamma_3bar, m_gamma_3bar, "3bar_equiv_20m", length = 20.0)

```

However the plot, see the example of the 1 bar equivalent in fig. 20.

UPDATE: the reasons for the plot being broken are explained in 3.3 and in short have to do with the units of different products being wrong, namely of Γ , L and (in a way) q . They have been fixed in the code above. The broken plot has been moved to the mentioned section.

With the code now fixed, we can finally look at the correct axion conversion probability plots. First for the 1 bar equivalent in fig. 6 and for the 3 bar equivalent in fig. 7. In the appendix in fig. 21 and 22 are the same plots for the full IAXO length of 20 m.

Although we still have the more pronounced cos behavior in our conversion probability, at least the widths of the peaks seem to match, as far as one can extract by eye from fig. 5.

In order to be able to determine the FWHM of the peaks (if we calculate the above curves for many pressure values) we can't simply fit a simple function to it and extract some sigma. While we can try to fit a simple gaussian, I don't expect it to fit very well, given the function of the conversion probability eq. 14.

Let's try it anyways for lack of a better way for now. We'll add it to the `genAxionConvPlot` proc above.




Figure 6: Example of the axion conversion probability at a pressure of $P = 14.3345$ mbar (corresponds to 1 bar at room temperature). This roughly reproduces the plot of fig. 5, although we see the influence of the \cos term a lot more. The only difference is that this plot corresponds to the BabyIAXO length of $L = 10$ m instead fig. 5 $L = 20$ m. For our IAXO equivalent plot see fig. 21 in the appendix. The absolute value of the probability is arbitrary, since $g_{a\gamma} = 1$ for this plot.




Figure 7: Example of the axion conversion probability at a pressure of $P = 43.0034 \text{ mbar}$ (corresponds to 3 bar at room temperature) and a magnet length of 10 m. For the IAXO equivalent plot see fig. 22 in the appendix.


```

1 import mpfit
2 proc gaussFit(p_ar: seq[float], x: float): float =
3   result = p_ar[0] * gauss(x = x, mean = p_ar[1], sigma = p_ar[2])
4
5 proc calcConvProbCurve(gamma, m_gamma, pressure: float, length = 10.0,
6   massRange = none[tuple[low, high: float]]()):
7   tuple[m_a, prob: seq[float]] =
8   var m_a: seq[float]
9   if massRange.isNone:
10    if pressure < 0.01:
11      echo "PRressure ", pressure
12      m_a = linspace(m_gamma * 0.2, m_gamma * 1.8, 1000)
13    else:
14      m_a = linspace(m_gamma * 0.5, m_gamma * 1.5, 1000)
15  else:
16    m_a = linspace(massRange.get.low, massRange.get.high, 1000)
17  let qs = m_a.mapIt(momentumTransfer(m_gamma, it))
18  let prob = m_a.mapIt(axionConversionProb(it, m_gamma, gamma, length = length))
19  result = (m_a: m_a, prob: prob)
20
21 proc fitToConvProb(gamma, m_gamma, pressure: float, nameSuffix = "",
22   length = 10.0, createPlot = true): (float, seq[float]) =
23   ## generates a single axion conversion probabilit plot
24   let (m_a, prob) = calcConvProbCurve(gamma, m_gamma, pressure, length)
25   let p0 = @[prob.max, m_a[prob.argmax], 0.02]
26   let (pRes, res) = fit(gaussFit, p0, m_a, prob, prob.mapIt(1.0))
27   #echoResult(pRes, res = res)
28   let probFit = m_a.mapIt(gaussFit(pRes, it))
29   if createPlot:
30     let df = seqsToDf({ "m_a / eV" : m_a,
31       "P_a->gamma" : prob,
32       "P_fit" : probFit })
33     #echo df
34     ggplot(df, aes("m_a / eV", "P_a->gamma")) +
35       geom_line() +
36       geom_line(aes(y = "P_fit", color = "Gaussian fit")) +
37       ggtitle("&"Gaussian fit to P_a->gamma at p = {pressure:.4f} mbar") +
38       ggsave("&"conv_prob_fit_{nameSuffix}.pdf")
39   result = (pressure, pRes)
40
41 let (drop, pRes1bar) = fitToConvProb(gamma_1bar, m_gamma_1bar, p1bar, "1bar")

```

If we ignore the code duplication we have right now **cough**, we can see from fig. 8 that a gaussian sort of works, at least for the purpose of extracting a reasonable σ . But let's not talk about our χ^2/dof please (hint $\sim 6.21 \times 10^{28}$).

In order to extract a FWHM, we simply have to take

$$\text{FWMH} = 2\sqrt{2\ln(2)}\sigma \approx 2.355\sigma \quad (27)$$

of our σ .

Let's do that.

```

1 proc fwhm(sigma: float): float =
2   ## returns the FWHM of a gaussian for a `sigma`
3   result = 2 * sqrt(2 * ln(2.0)) * abs(sigma)
4   echo "FWHM @ 1bar room temperature = ", fwhm(pRes1bar[2])

```

So the FWHM of the fit in fig. 8 is:

$$\text{FWHM}_{@1 \text{ bar}} \approx 0.001691 \text{ eV} \quad (28)$$




Figure 8: Gaussian fit applied to the axion conversion probability. While the χ^2 is beyond horrible, the fit is reasonable to extract the FWHM.

Thus, finally get back to our `logspace` derivation of some pressures, calculate $P_{a \rightarrow \gamma}$ for each, apply the fit and see where it leads us.

```

1 let pressuresFine = linspace(-6.0, 2.0, 1000)
2 let gammasFine = pressuresFine.mapIt(attenuationLength(it))
3 let mgammasFine = pressuresFine.mapIt(babyIaxoEffMass(it))
4 var fwhmFine: seq[float]
5 if false:
6   # NOTE: set to `true` if you want to recreate the fits and plots!
7   for i in 0 ..< pressuresFine.len:
8     let (p, pResI) = fitToConvProb(gammasFine[i], mgammasFine[i],
9                                   pressuresFine[i], &"{pressuresFine[i]:.6f}",
10                                   createPlot = true)
11     fwhmFine.add fwhm(pResI[2])

```

and now create the plot relating the pressure to the FWHM:

```

1 if false:
2   # NOTE: set to `true` if you want to redo these calculations
3   var dfFwhm = seqsToDf({ "Pressure / mbar" : pressuresFine,
4                           "FWHM / eV" : fwhmFine })
5   dfFwhm = dfFwhm.mutate(f{"FWHM / eV" ~ abs("FWHM / eV")})

```

```

6  ggplot(dfFwhm, aes("Pressure / mbar", "FWHM / eV")) +
7  geom_point() +
8  ylab(margin = 1.5) +
9  scale_y_log10() +
10 scale_x_log10() +
11 ggtitle("FWHM / eV of a->gamma probability depending on pressure / mbar") +
12 ggsave("fwhm_vs_pressure_full_loglog.pdf")
13 # and a subset filtered to above 0.01 mbar in linear
14 dfFwhm = dfFwhm.filter(f{"Pressure / mbar" > 0.01})
15 ggplot(dfFwhm, aes("Pressure / mbar", "FWHM / eV")) +
16 geom_point() +
17 scale_y_log10() +
18 ggtitle("FWHM / eV of a->gamma probability depending on pressure / mbar") +
19 ggsave("fwhm_vs_pressure_ylog.pdf")
20
21 let elements = pressuresFine.len
22 doAssert pressuresFine.len == elements
23 doAssert gammasFine.len == elements, " was " & $gammas.len
24 doAssert mgammasFine.len == elements, " was " & $mgammas.len
25 doAssert fwhmFine.len == elements, " was " & $fwhmFine.len
26
27 # finally write the data to a file
28 proc writeCsv(pressures, gammas, mgammas, fwhm: seq[float])
29 if false:
30   writeCsv(pressuresFine, gammasFine, mgammasFine, fwhmFine)

```

This gives us two plots. First in fig. 9 the full range from 1×10^{-6} mbar to 1×10^2 mbar in log-log. We see that below $\sim 1 \times 10^{-4}$ mbar the behavior changes significantly and probably the fit starts to break down. But in those conditions we should be well in the coherent part of the phase space anyways. So take that with a grain of salt. Keep in mind that these values are the cryogenic pressures, so ~ 43 mbar already equate to roughly 3 bar equivalent pressure at room temperature!

Then in fig. 10 we see the same data zoomed to all values above 0.01 mbar. Here we see the curve changes even faster than an exponential. Given the conversion probability itself already decays as an $\exp^{-\Gamma L}$ and the additional suppression of q in the denominator, this is to be expected I suppose. Or maybe not quite? Since for one plot, i.e. one value in the FWHM plots the only changing values is q . So it's just a $\frac{1}{x^2}$ dependency? Well, if I had more time, we could fit, but not today...

For once I'm not going to append **all** of the gaussian fits to the conversion probabilities. Otherwise this will have $\mathcal{O}(1000)$ pages...

One example of a very low pressure scenario is shown in fig. 11.

And let's implement the `writeCsv` proc to store the results.

```

1  proc writeCsv(pressures, gammas, mgammas, fwhm: seq[float]) =
2  var f = open("fwhm_results.csv", fmWrite)
3  f.write("# Pressures / mbar\t \Gamma / eV\t m_ / eV & fwhm / eV\n")
4  let elements = pressures.len
5  var line = ""
6  for i in 0 ..< elements:
7    line = &"{pressures[i]},{gammas[i]},{mgammas[i]},{abs(fwhm[i])}\n"
8    f.write(line)
9  f.close()

```




Figure 9: FWHM in eV of the axion conversion probability $P_{a \rightarrow \gamma}$ depending on the ^4He pressure inside BabyIAXO in mbar. The full range from 1×10^{-6} mbar to 1×10^2 mbar at cryogenic temperatures is shown. Below $\sim 1 \times 10^{-4}$ mbar the fit probably breaks.




Figure 10: FWHM in eV of the axion conversion probability $P_{a \rightarrow \gamma}$ depending on the ^4He pressure inside BabyIAXO in mbar. The range is zoomed to 1×10^{-2} mbar to 1×10^2 mbar at cryogenic temperatures. The curve seems to behave well in this range.




Figure 11: Example at which the gaussian fit fails at very low pressure. Even further below it's just a straight line with the dip.

2.4 Find our error in the calculation

It seems like calculating from the alternative m_γ calculation and putting in the pressure (derived by ideal gas law) equivalent of 1 bar at room temperature (yields 14.33 mbar at 4.2 K), matches with our expectation from the gas study, $m_\gamma \approx 0.261$ eV.

```

1 echo "Room temp: ", pressures.mapIt(pressureAtTemp(it) / 1000.0)
2
3 proc m_gamma_alternative(p: float): float =
4   # gallard alternative m_gamma. Says 0.02, more exact is:
5   result = sqrt(0.01988 * p / 4.2)
6
7 let compPress = pressureAtTemp(1000, 293.0, 4.2)
8 echo "cryo pressure for 1 bar at room: ", compPress
9 # get m_a at compPress
10 echo "m_a @ 1 bar @ room temp: ", babyIaxoEffMass(compPress)
11 echo "m_a alt. @ 1 bar room temp ", m_gamma_alternative(compPress)
12 doAssert babyIaxoEffMass(compPress).almostEqual(m_gamma_alternative(compPress))

```

$$PV = nRT \quad (29)$$

$$n = \frac{m}{M} \quad (30)$$

$$PV = \frac{m}{M}RT \quad (31)$$

$$P = \frac{m}{V} \frac{R}{M} T \quad (32)$$

$$P = \rho \frac{R}{M} T \quad (33)$$

$$\rho = \frac{PM}{RT} \quad (34)$$

And we know that these two are equivalent:

$$n = N_A c \quad (35)$$

$$n = \frac{N_A}{M} \rho \quad (36)$$

So this is the number density of molecules **molecules** with molar mass M . That means the number of **electrons** is Z times larger! That's where the factor of Z in equation 60 comes from!:

$$n_e = Z \frac{N_A}{M} \rho \quad (37)$$

Then we can alternatively express this in terms of the pressure instead of density, via

$$n_e = Z \frac{N_A}{M} \frac{PM}{RT} \quad (38)$$

$$n_e = Z \frac{N_A P}{RT} \quad (39)$$

$$(40)$$

which we can finally insert into the expression for m_γ eq. 4 to get:

$$m_\gamma = \sqrt{\frac{4\pi\alpha Z n_e P}{RT m_e}} \quad (41)$$

which after fixing the units and making P be in mbar:

$$m_\gamma = \sqrt{\frac{4\pi\alpha Z n_e (1.97 \times 10^{-7})^3 \cdot 100 \cdot P(\text{mbar})}{RT m_e}} \quad (42)$$

due to $100 \text{ Pa} = 1 \text{ mbar}$. Inserting all known constants:

$$m_\gamma = \sqrt{\frac{0.01988 \cdot P(\text{mbar})}{T(\text{K})}} \quad (43)$$

which is precisely equation 59.

Now let's make sure we can similarly derive the equation from Theodoros' thesis by inserting 60 instead:

$$m_\gamma = \sqrt{\frac{4\pi\alpha n_e}{m_e}} \quad (44)$$

$$m_\gamma = \sqrt{\frac{4\pi\alpha Z N_A \rho}{M m_e}} \quad (45)$$

Let's again consider the units of this:

$$\text{eV} = \sqrt{\frac{\text{kg} \cdot \text{mol}}{\text{m}^3 \cdot \text{g} \cdot \text{eV}}}, \quad (46)$$

based on the fact that we typically give the molar mass in g mol^{-1} . Yet, for the NIST data about the mass attenuation coefficient, it is more convenient, if we give our density not in kg/m^3 , but rather in g/cm^3 , because the attenuation coefficients are also given in cm^2/g . Which means we end up with the following corrections:

$$m_\gamma = \sqrt{\frac{4\pi\alpha Z N_A \cdot 1000 \cdot 1000 \cdot (1.97 \times 10^{-7})^3 \rho}{M m_e}} \quad (47)$$

$$m_\gamma = \sqrt{\frac{826.571 Z \rho}{M}} \quad (48)$$

$$m_\gamma = 28.75 \sqrt{\frac{Z}{M} \rho [\text{g/cm}^3]} \quad (49)$$

where one of the factors of 1000 is from the conversion of M to g mol^{-1} and the other from the conversion of ρ to g/cm^3 .

This means we have finally proven the seemingly arbitrary relation of m_γ .

TL;DR:

- we missed the factor Z in our initial calculation of the electron number density 7, because we forgot that there's Z electrons for each molecule apparently :P
- while trying to derive the equation for eq. 59 we were still missing a conversion from kg/m^3 to g/cm^3 for the density.

2.5 Absorption of X-rays in helium

In general the absorption of X-rays in a medium is also governed by the Beer-Lambert law, which for a uniform medium can be expressed via the attenuation length or the mass attenuation coefficient as:

$$I(d) = I_0 \exp\left(-\frac{\mu}{\rho} \rho d\right) = I_0 \exp(-\Gamma d) \quad (50)$$

where $\frac{\mu}{\rho}$ is again the mass attenuation coefficient and Γ the inverse absorption length.

The former expression is preferable, since we already have the values from tab. 1 and the interpolation eq. 16.

So for a given photon energy E we can calculate the X-ray flux after a traversed distance d :

```

1 proc intensitySuppression(energy, distance, pressure: float, temp = 293.15): float =
2   ## calculates the suppression factor of the intensity of X-rays
3   ## with `energy` (in keV) after `distance` (in `m`) at a helium `pressure`
4   ## (in `mbar`) and temperature of the gas of `temp` (in `K`).
5   ## By default assume room temperature for calc of beam line filled with

```



```

6  ## gas.
7  let massAtt = exp(logMassAttenuation(energy))
8  # calculate density from pressure
9  let rho = density(pressure, temp)
10 # factor 100 is to convert `distance` to `cm` from `m`
11 result = exp(-massAtt * rho * distance * 100)

```

2.5.1 Absorption in beam line at room temp @ 1 bar pressure

Assuming the BabyIAXO magnet is supposed to be filled with helium gas at room temperature (no cryo for the bore and even heated to 293.15 K), then we need to operate at pressures of $\mathcal{O}(1)$ bar). For a simple (and thin!) window setup it could be a nice solution to consider filling the beamline (incl. telescope) itself with helium at the same conditions. However, assuming a length of 5 m this results in a suppression of the intensity.

Let's calculate how large that suppression is and check how much intensity is lost after 5 m at 1000 mbar and 293.15 K for all energies 0 keV to 10 keV:

```

1  proc calcFilledBeamline(length = 5.0): DataFrame =
2  ## calculates the X-ray suppression of a beamline filled with gas at
3  ## `1 bar`, `293.15 K` of `5 m` (default) length
4  # take intensity suppression as is, assumes incoming intensity is `1`
5  let p = 1000.0 # mbar
6  let intensities = energies.mapIt(intensitySuppression(it, length, p))
7  result = seqsToDf({ "E / keV" : energies,
8                    "Transmission" : intensities })
9  ggplot(result, aes("E / keV", "Transmission")) +
10    geom_line() +
11    ggtitle(&"Transmission after {length} m of He at 1 bar, 293.15 K") +
12    ggsave(&"transmission_beamline_he_{length}_m.pdf")
13
14 let df5m = calcFilledBeamline(5.0)
15 let df7_5m = calcFilledBeamline(7.5)
16 echo "\n\n\n"
17 echo df5m.tail(10)
18 echo df7_5m.tail(10)
19 echo "done\n\n\n"

```

The same plot is also generated for 7.5 m. The plot for transmission after 5 m is shown in fig. 12.

Finally we wish to create a combined plot of the transmissions of the 5 m transmission, the 7.5 m case, a potential 10 μ m polypropylene (C_3H_6) window and the axion electron flux.

The transmission of the polypropylene window is calculated from http://henke.lbl.gov/optical_constants/filter2.html and the data file is polypropylene_window_10micron.txt.

```

1  proc transmissionsPlusSpectrum =
2  ## creates a plot of the expected transmissions for filled beamlines,
3  ## the polypropylene window and the axion electron spectrum.
4  let polypropDf = toDf(readCsv("polypropylene_window_10micron.txt", sep = ' ', skipLines = 1)).
5    mutate(f{"E / keV" ~ "Energy/eV" / 1000.0})
6  let gaeDf = toDf(readCsv("axion_gae_flux.dat", sep = ' ', skipLines = 9)).
7    rename(f{"E / keV" ~ "Energy/keV"}).
8    mutate(f{"Phi" ~ "Phi" / max("Phi") * 100.0}).
9    filter(f{"E / keV" <= 10.0})
10 # now generate a plot of E vs vals for each of the 4 datasets as one.
11 let fullDf = bind_rows([("Axion-Electron flux", gaeDf),

```




Figure 12: Transmission of X-rays after 5 m of ^4He at $P = 1000$ mbar and $T = 293.15$ K.

```

12         ("10  $\mu\text{m}$  PP", polypropDf),
13         ("5 m He", df5m),
14         ("7.5 m He", df7_5m),],
15         id = "Setup").
16     select("Setup", "E / keV", "Phi", "Transmission").
17     mutate(f{"Transmission" ~ "Transmission" * 100.0})
18     ggplot(fullDf.filter(f{"Setup" != "Axion-Electron flux"}), aes(x = "E / keV", y = "Transmission"))
19     geom_line(aes(color = "Setup")) +
20     geom_line(aes(x = "E / keV", y = "Phi"),
21               data = fullDf.filter(f{"Setup" == "Axion-Electron flux"})) +
22     scale_y_continuous("Transmission / %", secAxis = secAxis(name = "Axion flux / a.u.")) +
23     legend_position(x = 0.88, y = 0.1) +
24     ggtitle("Comparison of He filled beamline (1 bar, 293 K) and 10  $\mu\text{m}$  window") +
25     ggsave("window_he_transmissions_axion_flux.pdf", width = 853, height = 480)
26
27     transmissionsPlusSpectrum()

```

The resulting combined plot is shown in fig. 13.

2.6 First pressure value for He filling

The next thing to consider is to calculate the first pressure (or rather density) setting, which would be taken on a ^4He run. That is, from the "2.2", that is the pressure (density) equivalent for $m_a = 0.0228$ eV, that pressure is sought, which results in a FWHM of the same value as the declining vacuum sensitivity at that mass.

So we have to calculate two things:

1. the full vacuum sensitivity curve
2. determine the density (pressure) that corresponds to the m_γ , which produces the resonance curve (cf. fig. 7) that lines up with the vacuum curve at its FWHM on the left edge.




Figure 13: Comparison of three different possible setups. Either the beamline is filled with ^4He at $P = 1$ bar, $T = 293.15$ K and has a length of 5 m (red) or 7.5 m (green) or the beamline is under vacuum and a $10\text{ }\mu\text{m}$ C_3H_6 polypropylene window (blue) is used. In addition the expected solar axion-electron flux is shown in black in arbitrary units.

Let's start with 1.

2.6.1 Full vacuum sensitivity curve

Analytically the vacuum conversion probability can be derived from the expression eq. 14 by simplifying q for $m_\gamma \rightarrow 0$ and $\Gamma = 0$:

$$P_{a \rightarrow \gamma, \text{vacuum}} = \left(\frac{g_{a\gamma} B}{2} \right)^2 \frac{1}{q^2} [1 + 1 - 2 \cos(qL)] \quad (51)$$

$$P_{a \rightarrow \gamma, \text{vacuum}} = \left(\frac{g_{a\gamma} B}{2} \right)^2 \frac{2}{q^2} [1 - \cos(qL)] \quad (52)$$

$$P_{a \rightarrow \gamma, \text{vacuum}} = \left(\frac{g_{a\gamma} B}{2} \right)^2 \frac{2}{q^2} \left[2 \sin^2 \left(\frac{qL}{2} \right) \right] \quad (53)$$

$$P_{a \rightarrow \gamma, \text{vacuum}} = (g_{a\gamma} B)^2 \frac{1}{q^2} \sin^2 \left(\frac{qL}{2} \right) \quad (54)$$

$$P_{a \rightarrow \gamma, \text{vacuum}} = \left(\frac{g_{a\gamma} BL}{2} \right)^2 \left(\frac{\sin \left(\frac{qL}{2} \right)}{\left(\frac{qL}{2} \right)} \right)^2 \quad (55)$$

$$P_{a \rightarrow \gamma, \text{vacuum}} = \left(\frac{g_{a\gamma} BL}{2} \right)^2 \left(\frac{\sin(\delta)}{\delta} \right)^2 \quad (56)$$

$$(57)$$

Let's implement this and see if it reproduces the known plot.

```

1 proc vacuumConversionProb(m_a: float, length = 10.0): float =
2   ## calculates the conversion probability in BabyIAXO for the given axion
3   ## mass `m_a`
4   # both `g_agamma` and `B` only scale the absolute value `P`, does not matter
5   const g_agamma = 1.0 #1e-11
6   const B = 4.5 # T, actually don't know the real number right now
7   # convert length in `m` to `eV`

```

```

8   let L = length / 1.97e-7 # m
9   let E_a = 4.2 # keV mean value for Primakoff spectrum
10  let q = momentumTransfer(m_gamma = 0.0, m_a = m_a)
11  let term1 = pow(g_agamma * B * L / 2.0, 2)
12  let term2 = pow(sin(q * L / 2.0) / (q * L / 2.0), 2.0)
13  result = term1 * term2
14
15  proc vacuumConversionProbApprox(m_a: float, length = 10.0): float =
16    ## approximates the vacuum conversion probability to second order of taylor
17    ## expansion (4th power of `sin^2` argument)
18    # both `g_agamma` and `B` only scale the absolute value `P`, does not matter
19    const g_agamma = 1.0 #1e-11
20    const B = 4.5 # T, actually don't know the real number right now
21    # convert length in `m` to `eV`
22    let L = length / 1.97e-7 # m
23    let E_a = 4.2 # keV mean value for Primakoff spectrum
24    let q = momentumTransfer(m_gamma = 0.0, m_a = m_a)
25    let term1 = pow(g_agamma * B * L / 2.0, 2)
26    let term2 = 1.0 - pow(q * L / 2.0, 2.0) / 3.0
27    result = term1 * term2
28
29  proc vacuumConvProbPlot(useApprox = false) =
30    ## generates the typical plot of the conversion probability in the range
31    ## 1 peV to 1 eV for g_a,gamma = 1.0 (for simplicity)
32    # we use linspace so we have more fidelity where the curve is more interesting
33    let vacMasses = linspace(1e-6, 1.0, 1000)
34    var probs: seq[float]
35    var outname = ""
36    if not useApprox:
37      probs = vacMasses.mapIt(vacuumConversionProb(it))
38      outname = "vacuum_conversion_prob.pdf"
39    else:
40      probs = vacMasses.mapIt(vacuumConversionProbApprox(it))
41      outname = "vacuum_conversion_prob_taylor.pdf"
42    let dfVac = seqsToDf({ "m_a / eV" : vacMasses,
43                          "P_a,gamma" : probs})
44
45    ggplot(dfVac, aes("m_a / eV", "P_a,gamma")) +
46      geom_line() +
47      #geom_point() +
48      scale_x_log10() +
49      scale_y_log10() +
50      ggsave(outname)
51  vacuumConvProbPlot()
52  vacuumConvProbPlot(useApprox = true)

```

The resulting vacuum conversion probability plot is shown in fig. 14.

2.6.2 Determine the next pressure (density) step

Now starts the more complicated part of the calculation, namely 2.

It can be decomposed into a few parts:

- the vacuum curve
- a set of (m_γ, Γ, ρ) , which defined the conversion probability with a buffer gas
- calculate the mass and conversion probability at the left edge of the FWHM of the buffer curve




Figure 14: Normal axion photon conversion probability in the BabyIAXO magnet assuming $g_{a,\gamma} = 1$ in the typical mass range. The behavior for higher masses is due to being off-resonance ($qL > \pi$) and is usually not shown in the vacuum plots.

- find m_a of the vacuum curve for the P of the (m_a, P) point of the buffer curve
- calculate the distance mass distance between $m_{a,\text{vacuum}}$ at P and $m_{a,\text{left}}$ FWHM
- take absolute value of distance
- perform minimization of that distance. Minimum corresponds to desired density (pressure) point.

Let's build this up in code then.

```

1  proc pressureToGammaMGamma(pressure: float): tuple[gamma, m_gamma: float] =
2      ## returns a tuple of the corresponding `gamma` and
3      ## `m_gamma` values associated with the pressure for BabyIAXO filled with
4      ## He.
5      result = (gamma: attenuationLength(pressure),
6                m_gamma: babyIaxoEffMass(pressure))
7
8  proc getDfVac(massRange: Option[tuple[low, high: float]]): DataFrame =
9      let r = if massRange.isNone: (low: 1e-6, high: 1.0) else: massRange.get
10     let vacMasses = linspace(r.low, r.high, 1000)
11     let vacProbs = vacMasses.mapIt(vacuumConversionProb(it))
12     result = seqsToDf({ "m_a / eV" : vacMasses,
13                        "P_a->gamma" : vacProbs})
14     # echo result.pretty(50)
15
16  type
17      MassFit = object
18      discard
19  proc vacuumMassGivenProb(prob: float): float =
20      ## determines the mass corresponding to the given probability.
21      ## Since `P` even for the vacuum case is not really invertible,

```

```

22     ## we'll do it numerically.
23     proc probDiff(p0: seq[float], dummy: MassFit): float =
24         result = abs(vacuumConversionProb(p0[0]) - prob)
25     var opt = newNloptOpt[MassFit](LN_COBYLA, 1)
26     # hand the function to fit as well as the data object we need in it
27     var varStr = newVarStruct(probDiff, MassFit())
28     opt.setFunction(varStr)
29     let (params, minVal) = opt.optimize(@[0.01])
30     result = params[0]
31     opt.destroy()
32     echo vacuumMassGivenProb(1.5e15)
33
34     proc pressureGivenEffPhotonMass(m_gamma: float, T = 4.2): float =
35         ## calculates the inverse of `babyIaxoEffPhotonMass`, i.e. the pressure
36         ## from a given effective photon mass in BabyIAXO
37         result = m_gamma * m_gamma * T / 0.01988
38
39     template fwhmTail(name: untyped, op: untyped): untyped =
40         proc `name` (m_a: float, outputInfo = false,
41             massRange = none[tuple[low, high: float]]()):
42             tuple[m_as, probs: seq[float], mAtFwhm, pHalf: float] =
43                 # calculate pressure corresponding to `m_a`
44                 let pressure = pressureGivenEffPhotonMass(m_a)
45                 # get gamma and m_gamma from pressure
46                 let (gamma, m_gamma) = pressureToGammaMGamma(pressure)
47                 doAssert abs(m_a - m_gamma) < 1e-4
48                 let (m_as, probs) = calcConvProbCurve(gamma, m_gamma,
49                     pressure, massRange = massRange)
50                 let (p, pResI) = fitToConvProb(gamma, m_gamma, pressure,
51                     nameSuffix = "teststest", createPlot = true)
52                 let fwhmVal = fwhm(pResI[2])
53                 let mAtFwhm = `op` (pResI[1], fwhmVal / 2.0)
54                 let pHalf = pResI[0] / 2.0
55                 result = (m_as: m_as, probs: probs, mAtFwhm: mAtFwhm, pHalf: pHalf)
56                 if outputInfo:
57                     echo "Axion mass m_a = ", m_a, " eV"
58                     echo "Pressure P = ", pressure, " mbar"
59                     echo "Attenuation length Gamma = ", gamma
60                     echo "Effective photon mass m_gamma = ", m_gamma, " eV"
61                     echo "FWHM ", fwhmVal
62
63     fwhmTail(lhsFwhmTail, `~`)
64     fwhmTail(rhsFwhmTail, `+`)
65
66     proc massDifference(m_a, prob: float, cmpMass = none[float]()): float =
67         ## Given a conversion probability value at a certain mass, calculate the
68         ## mass difference between this value and the vacuum curve
69         var ma_cmp: float
70         if cmpMass.isNone:
71             ma_cmp = vacuumMassGivenProb(prob)
72         else:
73             let (m_as, probs, mAtFwhm, pHalf) = rhsFwhmTail(cmpMass.get)
74             ma_cmp = mAtFwhm
75             echo "using ", cmpMass
76             echo "Corresponding ma in vacuum ", ma_cmp, " for ", prob
77             result = abs(m_a - ma_cmp)
78
79     proc gasStepsPlot(massLhs, massRhs: tuple[name: string, m: float], title = "",

```

```

80         filterMass = none[tuple[low, high: float]]() =
81     ## generates a plot showing the resonance curves for the two masses
82     ## (or one if one mass is 0.0). Compares it to the vacuum conversion
83     ## probability.
84     let (m_as, probs, mAtFwhm, pHalf) = lhsFwhmTail(massLhs.m, massRange = filterMass)
85     let dfLhs = seqsToDf({ "m_a / eV" : m_as,
86                           "P_a->gamma" : probs,
87     })
88     echo "LHS = ", massLhs
89     echo "LHS P = ", pressureGivenEffPhotonMass(massLhs.m)
90     var dfComb: DataFrame
91     if massRhs.name.len == 0:
92         dfComb = bind_rows(["vacuum", getDfVac(filterMass)),
93                             (massLhs.name, dfLhs)],
94                             #("xline", dfDummy),
95                             #("yline", dfDummy2),
96                             id = "case")
97     else:
98         echo "RHS = ", massRhs
99         echo "RHS P = ", pressureGivenEffPhotonMass(massRhs.m)
100        let (m_Rhs, probsRhs, , _) = rhsFwhmTail(massRhs.m, massRange = filterMass)
101        let dfRhs = seqsToDf({ "m_a / eV" : m_Rhs,
102                                "P_a->gamma" : probsRhs,
103        })
104        dfComb = bind_rows(["vacuum", getDfVac(filterMass)),
105                            (massRhs.name, dfRhs),
106                            (massLhs.name, dfLhs)],
107                            id = "case")
108    if filterMass.isSome:
109        let fm = filterMass.get
110        dfComb = dfComb.filter(f{"m_a / eV" >= fm.low and "m_a / eV" <= fm.high})
111    var plt = ggplot(dfComb, aes("m_a / eV", "P_a->gamma", color = "case")) +
112        geom_line() +
113        scale_y_log10() +
114        ggtitle(title)
115    if filterMass.isNone:
116        plt = plt + scale_x_log10()
117    let n1 = massLhs.name.replace(" ", "_")
118    let n2 = massRhs.name.replace(" ", "_")
119    plt + ggsave(&"vacuum_helium_cutoff_{mAtFWHM}_{n1}_{n2}.pdf")
120
121
122    proc massDifferenceAtMass(m_a: float, createPlot = true,
123        outputInfo = false, cmpMass = none[float]()): float =
124    ## returns the pressure that corresponds to the mass that is required to
125    ## achieve a FWHM overlap between the vacuum curve and the buffer gas setup
126    # use `m_a` as start
127    # calculate buffer curve with this value
128    echo "Eff photon mass ", m_a
129    let (m_as, probs, mAtFwhm, pHalf) = lhsFwhmTail(m_a, outputInfo)
130    # find m_a of `pHalf` on vacuum curve
131    result = massDifference(mAtFwhm, pHalf, cmpMass)
132    if createPlot:
133        if cmpMass.isNone:
134            gasStepsPlot((name: "helium", m: m_a), massRhs = (name: "", m: 0.0))
135        else:
136            gasStepsPlot((name: "helium", m: m_a),
137                (name: "heliumRef", m: cmpMass.get))

```

```

138
139 proc findMassAtFwhm(m_a: float, cmpWithMassless = true): float =
140   ## performs minimization of `massDifferenceAtMass` to find the mass `m_a` at which
141   ## the FWHM of the buffer phase matches the vacuum line
142   proc findMin(p0: seq[float], dummy: MassFit): float =
143     if cmpWithMassless:
144       result = massDifferenceAtMass(p0[0])
145     else:
146       result = massDifferenceAtMass(p0[0], cmpMass = some(m_a))
147
148   var opt = newNloptOpt[MassFit](LN_COBYLA, 1, bounds = @[1e-4, Inf])
149   # hand the function to fit as well as the data object we need in it
150   var varStr = newVarStruct(findMin, MassFit())
151   opt.setFunction(varStr)
152   opt.xtol_rel = 1e-8
153   opt.ftol_rel = 1e-8
154   let (params, minVal) = opt.optimize(@[m_a])
155   result = params[0]
156   opt.destroy()
157
158 let mstep1 = findMassAtFwhm(babyIaxoVacuumMassLimit)
159 let mstep2 = findMassAtFwhm(mstep1, cmpWithMassless = false)

```

Finally generate the plots from the calculation:

```

1 echo "\n\n-----\n\n"
2 proc toPressureDiff(m1, m2: float, T = 4.2): float
3 let pdiffVacStep1 = toPressureDiff(mstep1, babyIaxoVacuumMassLimit)
4 gasStepsPlot((name: "1st He step", m: mstep1),
5               massRhs = (name: "", m: 0.0),
6               title = &"First buffer gas step; match at FWHM,  $\Delta P = \{pdiffVacStep1:.4f\}$  mbar (@ 4.2 K)
7 let pdiffFirst2Steps = toPressureDiff(mstep1, mstep2)
8 gasStepsPlot((name: "2nd He step", m: mstep2),
9               (name: "1st He step", m: mstep1),
10              title = &"First two He steps, each match at FWHM,  $\Delta P = \{pdiffFirst2Steps:.4f\}$  mbar (@ 4

```

The first gas buffer step is shown in figure 15. We can see that we get the expected matching of the first coherent curve with the dropping vacuum curve.

The first two helium steps are then shown in fig. 16. Again we see that the matching of the two coherent curves is correct.

At 4.2 K the first pressure step is at 0.1927 mbar and the second at 0.3808 mbar, meaning a difference in pressure of 0.1881 mbar.

Now let's look at the same for the steps between two pressures at 1 bar equivalent and 3 bar equivalent. For this calculate the mass that corresponds to these and take values for `findMassAtFwhm`.

To better present the information, let's write a helper proc, which converts a difference in masses to a difference in pressures:

```

1 proc toPressureDiff(m1, m2: float, T = 4.2): float =
2   ## assuming two masses `m1` and `m2` in `eV`, converts the
3   ## mass difference to a corresponding pressure difference
4   ## at temperature `T`
5   result = abs(pressureGivenEffPhotonMass(m2, T = T) -
6               pressureGivenEffPhotonMass(m1, T = T))

```





Figure 15: Plot of the vacuum conversion probability depending on the axion mass for the vacuum case and the first pressure step with ^4He buffer gas, such that the two produce a "best matching", i.e. the sensitivity does not drop below the value at the FWHM (= by a factor of 2 in probability). At 4.2 K the pressure is 0.1927 mbar.




Figure 16: Plot of the vacuum conversion probability depending on the axion mass for the vacuum case and the first two pressure steps with ^4He buffer gas. We can see the mass difference achieved for the first two steps. At 4.2 K the first pressure step is at 0.1927 mbar and the second at 0.3808 mbar.

```

1 # 1 bar
2 let mstep1Bar = findMassAtFwhm(m_gamma_1bar, cmpWithMassless = false)
3 let pdiff1bar = toPressureDiff(m_gamma_1bar, mstep1Bar)
4 gasStepsPlot((name: "Step 1", m: m_gamma_1_bar),
5              (name: "Step 2", m: mstep1Bar),
6              title = &"Two steps at 1 bar (@ 293 K) pressure, ΔP = {pdiff1bar:.4f} mbar (@ 4.2 K)"
7              filterMass = some((low: 0.22, high: 0.2999)))
8 #
9 ### 3 bar
10 let mstep3Bar = findMassAtFwhm(m_gamma_3bar, cmpWithMassless = false)
11 let pdiff3bar = toPressureDiff(m_gamma_3bar, mstep3Bar)
12 gasStepsPlot((name: "Step 1", m: m_gamma_3_bar),
13              (name: "Step 2", m: mstep3Bar),
14              title = &"Two steps at 3 bar (@ 293 K) pressure, ΔP = {pdiff3bar:.4f} mbar (@ 4.2 K)"
15              filterMass = some((low: 0.42, high: 0.48)))

```

The step difference near 1 bar at $T = 293.15$ K is shown in fig. 17. Here the pressure difference at cryo temperature 4.2 K is only




Figure 17: Plot of the vacuum conversion probability depending on the axion mass for the vacuum case and two pressure steps with ^4He buffer gas at 1 bar at $T = 4.2$ K.

And in fig. 18 we see the same plot near the 3 bar pressure equivalent.

Finally we still want to reproduce fig. 3 depending on the density instead of a pressure.

```

1 proc m_a_vs_density(pstart, pstop: float) =
2   let pressures = logspace(pstart.log10, pstop.log10, 1000)
3   let densities = pressures.mapIt(density(it, 4.2))
4   let masses = pressures.mapIt(babyIaxoEffMass(it)) # corresponding masses

```




Figure 18: Plot of the vacuum conversion probability depending on the axion mass for the vacuum case and two pressure steps with ^4He buffer gas at 3 bar at $T = 4.2$ K.

```

5 # convert both seqs to a dataframe
6 let ref1Bar = density(1000, 293.15)
7 let df1bar = seqsToDf({" / g/cm^3" : @[ref1Bar, ref1Bar], "m_a / eV" : @[1e-2, 1.0]})
8 let ref3Bar = density(3000, 293.15)
9 let df3bar = seqsToDf({" / g/cm^3" : @[ref3Bar, ref3Bar], "m_a / eV" : @[1e-2, 1.0]})
10 let refVacLimit = density(pressureGivenEffPhotonMass(babyIaxoVacuumMassLimit))
11 let dfVacLimit = seqsToDf({" / g/cm^3" : @[refVacLimit, refVacLimit], "m_a / eV" : @[1e-2, 1.0]})
12 let df = seqsToDf({" / g/cm^3" : densities, "m_a / eV" : masses})
13 let dfComb = bind_rows([("ma vs ", df),
14                           ("1 bar @ 293 K", df1bar),
15                           ("3 bar @ 293 K", df3bar),
16                           ("Vacuum limit", dfVacLimit)],
17                          id = "Legend")
18 ggplot(dfComb, aes(" / g/cm^3", "m_a / eV", color = "Legend")) +
19   geom_line() +
20   scale_x_log10() +
21   scale_y_log10() +
22   ggtitle("Sensitive axion mass in eV depending on helium density in g / cm^3") +
23   ggsave("m_a_vs_density.pdf")
24
25 m_a_vs_density(pressureGivenEffPhotonMass(babyIaxoVacuumMassLimit) * 0.9,
26               pressureGivenEffPhotonMass(mstep3Bar) * 1.1)

```

The final plot then is the dependency of the axion mass m_a against the ^4He density in the magnet. This is shown in fig. 19.




Figure 19: Sensitive axion mass m_a depending on the density of ^4He buffer gas in the magnet. Also added are the lines corresponding to the vacuum limit and the 1 bar and 3 bar lines (at 4.2 K).

3 Notes

3.1 TODO Evaluate if this m_γ results in the same value as our fn

According to Theodoros' PhD thesis: <https://cds.cern.ch/record/1607071/files/CERN-THESIS-2012-349.pdf> (page 186) the effective photon mass can also be written as:

$$m_\gamma = 28.77 \sqrt{\frac{Z}{A} \rho} \quad (58)$$

where Z is the atomic number, ρ the density of the gas and A the atomic mass of the buffer gas.

"Distinguishing axion models at IAXO, Jaeckel & Thompson:" <https://iopscience.iop.org/article/10.1088/1475-7516/2019/03/039/meta> mentions another approximation for m_γ , namely:

$$m_\gamma \approx \sqrt{0.02 \frac{p(\text{mbar})}{T(\text{K})}} \text{ eV} \quad (59)$$

which seems to be quite a bit closer to our assumption than Theodoros'.

J. Gallan's PhD thesis seems to be the original source for the equation, <https://arxiv.org/pdf/1102.1406.pdf>.

He arrives at it from the relation that the electron number density can be written as:

$$n_e = Z \frac{N_A}{M} \rho \quad (60)$$

I should take a pen and paper and write it down. Continue on with the conversion probability.

3.2 Momentum transfer units

Momentum transfer has units of momentum.

Momentum has natural units of:

3.3 Attenuation length, momentum transfer and length

This section was to attempt to fix the plot in fig. 20.




Figure 20: Example of the axion conversion probability at a pressure of $P = 14.3345$ mbar (corresponds to 1 bar at room temperature). This should reproduce fig. 5. That apparently does not work yet.

So in principle: $\exp -\Gamma L$ has to be unit 1, i.e. Γ has to be unit of inverse length (which is by definition true of course, since it's defined as an inverse length). However, in the units we actually use for our numbers, that ain't quite the case. We use SI units for $L = 10$ m and have Γ from:

$$\Gamma = \rho \left(\frac{\mu}{\rho} \right) \quad (61)$$

Now, we use ρ in units of g/cm^3 , while the NIST data supplies μ/ρ in cm^2/g . What does this mean? It means that

$$[\Gamma] = [\text{g}/\text{cm}^3] [\text{cm}^2/\text{g}] \quad (62)$$

$$[\Gamma] = [\text{cm}^{-1}] \quad (63)$$

ref eq 14.

This means two things for us. One: For all products of $\Gamma \cdot L$ we have to apply a correction factor of 100 to convert from cm to m, but maybe more importantly, we have to consider what this means for the denominator of the second term of the conversion probability, namely:

$$q^2 + \Gamma^2/4 \quad (64)$$

Do we have to fix one of these?

The same about the products holds also for the argument to the cosine:

$$\cos(q \cdot L) \quad (65)$$

Since we have by definition q in eV and




Figure 21: Example of the axion conversion probability at a pressure of $P = 14.3345$ mbar (corresponds to 1 bar at room temperature) for the full IAXO length 20 m. This is equivalent to the plot of fig. 5, although we see the influence of the cos term a lot more.




Figure 22: Example of the axion conversion probability at a pressure of $P = 43.0034 \text{ mbar}$ (corresponds to 3 bar at room temperature) and a magnet length of 20 m. This plot should be exactly the equivalent to the 3 bar plot of the IAXO gas phase study paper.